Pensieve header: Tristram-Levine signatures for braid closures. Continues pensieve://2021-02/.

In[ ]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\Signatures"];
<< Common.m
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

pdf

Tristram-Levine for Braid Closures

tex

{\def\nbpdfText#1{\vskip 1mm\par\noindent\includegraphics[width=\linewidth]{#1}}

pdf



tex

}

pdf

IN[*i*, *j*, *k*, *l*] measures the upper-half-plane intersection number of the arrows $i \to j$ and $k \to l$.
IN[1, 3, 2, 4] is +1.

pdf

In[ ]:=
```
χ[cond_] := If[TrueQ[cond], 1, 0];
IN[i_, j_, k_, l_] := χ[i > l] + χ[j > k] - χ[i > k] - χ[j > l]
```

pdf

SC stands for “Simple Cycle”.

pdf

```
In[•]:= SCs[β_BR] := Module[{n = β[[1]]}, Flatten@Table[
          SC @@@ Subsets[Flatten@Position[Abs[β[[2]]], k], {2}], {k, n - 1}] ]
```

```
In[•]:= Knot[4, 1] // BR // SCs
```

KnotTheory: The minimum braids representing the knots with up to 10 crossings were provided by Thomas Gittings. See arXiv:math.GT/0401051.

```
Out[•]= {SC[1, 3], SC[2, 4]}
```

pdf

lk[β, sc1, sc2] computes lk(sc1, sc2$^+$) in β.

pdf

```
In[•]:= lk[β_BR, SC[i_, j_], SC[k_, l_]] := Module[
          {n = β[[1]], s1 = Abs@β[[2, i]], s2 = Abs@β[[2, k]]},
          Which[
           s2 - s1 == 1, IN[i, j, l, k],
           s1 == s2, χ[β[[2, i]] > 0] (χ[i == l] - χ[i == k]) +
            χ[β[[2, j]] > 0] (χ[j == k] - χ[j == l]) - IN[i + 0.1, j + 0.1, k, l],
           True,
           0] ]
```

```
In[•]:= Block[{β = Knot[4, 1] // BR},
         Table[lk[β, sc1, sc2], {sc1, SCs[β]}, {sc2, SCs[β]}] // MatrixForm]
```

```
Out[•]//MatrixForm=
(  1  -1 )
(  0  -1 )
```

pdf

SM for Seifert Matrix.

pdf

```
In[•]:= SM[β_BR] := SM[β] = Module[{n = β[[1]], col, H1},
          H1 = Flatten@Table[
              col = Flatten@Position[Abs[β[[2]]], k];
              SC[First@col, #] & /@ Rest[col],
              {k, n - 1}];
          Table[lk[β, sc1, sc2], {sc1, H1}, {sc2, H1}] ];
```

```
In[•]:= SM[BR@Knot[6, 1]] // MatrixForm
```

```
Out[•]//MatrixForm=
(  1  0   0   0 )
(  1  0  -1   0 )
(  0  0   1  -1 )
(  0  0   0  -1 )
```

pdf

```
In[•]:= AlexFromSM[K_] := Module[{A = SM@BR@K}, Det[t A - A^T]]
```

```
In[•]:= Union@Table[Simplify[Alexander[K][t]/AlexFromSM[K]], {K, AllKnots[{3, 10}]}]
```

```
Out[•]= {1/t^5, 1/t^4, 1/t^3, 1/t^2, 1/t}
```

pdf

```
In[•]:= SignatureFromSM[K_] := Module[{A = SM@BR@K}, MatrixSignature[A + A^T]]
```

```
In[•]:= SignatureFromSM /@ AllKnots[{3, 9}]
```

```
Out[•]= {2, 0, 4, 2, 0, 2, 0, 6, 2, -4, -2, 4, 2, 0, 0, 4, 0, 2, -4, 2, -2, 0, 0, -2, 2, 0, 0, 2, 4, 2, 0,
 0, -6, 0, 2, 8, 2, -6, 4, -2, 6, 4, 2, 6, -4, -4, 2, -4, 0, -2, -6, 2, 4, 0, 4, -2, -2,
 4, 0, 2, -2, 0, 2, 2, 0, 2, -2, 0, 0, 2, -4, 0, 4, -2, 2, 0, -2, -4, 0, 2, 0, -2, -2, -4}
```

```
In[•]:= -KnotSignature /@ AllKnots[{3, 9}]
```

```
Out[•]= {2, 0, 4, 2, 0, 2, 0, 6, 2, -4, -2, 4, 2, 0, 0, 4, 0, 2, -4, 2, -2, 0, 0, -2, 2, 0, 0, 2, 4, 2, 0,
 0, -6, 0, 2, 8, 2, -6, 4, -2, 6, 4, 2, 6, -4, -4, 2, -4, 0, -2, -6, 2, 4, 0, 4, -2, -2,
 4, 0, 2, -2, 0, 2, 2, 0, 2, -2, 0, 0, 2, -4, 0, 4, -2, 2, 0, -2, -4, 0, 2, 0, -2, -2, -4}
```

pdf

```
In[•]:= TLSFromSM[K_, ω_] := Module[{A = SM@BR@K}, MatrixSignature[(1 - ω) A + (1 - ω*) A^T]]
```

tex

{\def\nbpdfOutput#1{\vskip 1mm\[\includegraphics[width=0.5\linewidth]{#1} \]}

pdf

```
In[•]:= Plot[TLSFromSM[Knot@"K12a422", e^(i t)], {t, 0, 2 π}]
```

pdf

Out[•]=



tex

}